

# mpitb: A toolbox for multidimensional poverty indices

Nicolai Suppa

Spanish Stata Conference  
Madrid, October 2022



**CED**  
Centre d'Estudis  
Demogràfics



Institució  
**CERCA**  
Centres de Recerca  
de Catalunya



**OPHI**  
Oxford Poverty & Human  
Development Initiative



**OXFORD**  
DEPARTMENT OF  
INTERNATIONAL  
DEVELOPMENT



UNIVERSITY OF  
**OXFORD**

- ① How to measure MDP?
- ② What `mpitb` can and cannot do
- ③ The basic syntax
- ④ Examples

How to measure MDP?

 Alkire and Foster (2011) and Alkire, Foster, et al. (2015)

## Identification (who is poor?)

- individuals or households  $i = 1, \dots, N$
- (binary) deprivation indicators  $j = 1, \dots, D$
- deprivation score  $c_i$  relying weights  $0 \leq w_j \leq 1$  with  $\sum w_j = 1$
- individual is poor if  $c_i \geq k$  (poverty cutoff)

## Aggregation (how much poverty?)

- headcount ratio  $H$
- intensity  $A$
- adjusted headcount ratio  $M = H \times A$
- deprivation rate  $h_j$  and censored deprivation rate:  $h_j(k)$

## What `mpitb` can and cannot do

mpitb can

- **estimate key quantities** for MDP analysis (incl. their SE)
- for **parameter sets** (e.g., cutoffs) and **subgroups** (e.g., regions)
- facilitate **cross-country** estimation and analysis.

Why care?

- 10 indicators, 11 cutoffs for national, urban-rural, 20 regions
- 8601 point estimates  $((33 \times 11 + 10) \times 23 + 22)$

Main benefits

- save time for programming and estimation
- efficient workflow with well-structured results, facilitates replication, bug tracking, cross-checks
- adopt commands from akin projects (e.g., graphs)

quantity	level	change <sup>1</sup>
adj. headcount ratio	●	●
headcount ratio	●	●
intensity	●	●
censored depr. rate	●	●
uncensored depr. rate	●	●
absolute ind. contr.	●	
percentage ind. contr.	●	
indicator non-response	●	
retained sample	●	

Notes: <sup>1</sup> may be reported as (i) absolute or relative, (ii) annualised or raw.

mpitb cannot

- generate deprivation indicators
- make normative decisions
- prevent estimation of wrong or meaningless numbers
- provide every number needed a particular research project



## The basic syntax

**mpitb set**

```
mpitb set [ , name(mpiname)
           d1(varlist, subopts) ... d5(varlist, subopts)
           description(text) clear replace ]
```

**mpitb est**

```
mpitb est , name(mpiname) [ ...
  klist(numlist) weights(wgts sopts)
  measures(mlist) indmeasures(imlist) aux(auxlist)
  lframe(name [, sopts]) lsave(filename [, sopts])
  over(varlist [, sopts])
  ... ]
```

# Examples

## Example 1: A single year for a single country

```
. use syn_cdta.dta if t == 1 , clear
```

```
. sum
```

Variable	Obs	Mean	Std. dev.	Min	Max
d_nutr	7,439	.2521844	.4342958	0	1
d_cm	7,500	.0629333	.2428592	0	1
d_satt	7,484	.3178781	.4656829	0	1
d_educ	7,500	.2993333	.4579966	0	1
d_elct	7,500	.3976	.4894346	0	1
d_sani	7,500	.2384	.4261334	0	1
d_wtr	7,500	.2737333	.4459035	0	1
d_hsg	7,500	.4177333	.4932186	0	1
d_ckfl	7,500	.1484	.3555197	0	1
d_asst	7,500	.2829333	.4504543	0	1
area	7,500	.5989333	.4901471	0	1
region	7,500	10.53347	5.808389	1	20
stratum	7,500	1055.853	580.8484	100	2005
psu	7,500	1055856	580848.3	100000	2005005
weight	7,500	1	0	1	1
year	7,500	2010	0	2010	2010
t	7,500	1	0	1	1

```
. svyset psu [pw=weight], strata(stratum)
Sampling weights: weight
                   VCE: linearized
                   Single unit: missing
                   Strata 1: stratum
Sampling unit 1: psu
                   FPC 1: <zero>

. mpitb set , na(trial01) d1(d_cm d_nutr, na(hl)) d2(d_satt d_educ, na(ed)) ///
>         d3(d_elct d_wtr d_sani d_hsg d_ckfl d_asst, name(ls)) de(pref. spec)
```

```
. mpitb est , name(trial01) meas(all) indmeas(all) aux(hd) klist(20 33 50) ///
> weight(equal) svy lfr(myresults, replace) over(region area)
```

---

Specification

Name: trial01.

Weighting scheme: equal.

Description: pref. spec

---

Dimension 1: hl	0.3333	(d_cm d_nutr)
Dimension 2: ed	0.3333	(d_satt d_educ)
Dimension 3: ls	0.3333	(d_elct d_wtr d_sani d_hsg d_ckfl d_asst)

---

Indicator 1: d_cm	0.1667
Indicator 2: d_nutr	0.1667
Indicator 3: d_satt	0.1667
Indicator 4: d_educ	0.1667
Indicator 5: d_elct	0.0556
Indicator 6: d_wtr	0.0556
Indicator 7: d_sani	0.0556
Indicator 8: d_hsg	0.0556
Indicator 9: d_ckfl	0.0556
Indicator 10: d_asst	0.0556

---

No missing indicator was found.

---

 Estimation
 

---

```
# accumulated estimates (levels): 19 (national main completed)
# accumulated estimates (levels): 109 (national indicators completed)
# accumulated estimates (levels): 489 (region completed)
# accumulated estimates (levels): 2289 (region indicators completed)
# accumulated estimates (levels): 2347 (area completed)
# accumulated estimates (levels): 2527 (area indicators completed)
(note: frame myresults not found)
```

---

 Result frames & files
 

---

Level frame (myresults): Estimates overview

Number of subgroups:

```
area:      2
region:    20
```

	level of analysis			Total
	area	nat	region	
measure				
A	6	3	60	69
H	6	3	60	69
MO	6	3	60	69
actb	60	30	600	690



hd	20	10	200	230
hdk	60	30	600	690
pctb	60	30	600	690
popsh	2		20	22
Total	220	109	2,200	2,529

---

Number of parameters:

k: 3 (20 33 50)  
 wgts: 1 (equal)  
 spec: 1 (trial01)

---

```
. cwf myresults
```

```
. d
```

```
Contains data
```

```
Observations:      2,529
```

```
Variables:         14
```

---

Variable name	Storage type	Display format	Value label	Variable label
b	float	%5.4f		point estimate
se	float	%5.4f		standard error
ll	float	%5.4f		CI lower bound
ul	float	%5.4f		CI upper bound
pval	float	%4.2f		p-value
tval	float	%4.2f		t-value
loa	str10	%10s		level of analysis
measure	str10	%10s		measure
indicator	str10	%10s		indicator
spec	str10	%10s		name of specification
wgts	str10	%10s		weighting scheme
k	float	%9.0g		poverty cutoff
ctype	byte	%8.0g	ctype	type of change
subg	int	%8.0g		subgroup

---

```
Sorted by:
```

```
Note: Dataset has changed since last saved.
```

```
. tab measure loa
```

measure	level of analysis			Total
	area	nat	region	
A	6	3	60	69
H	6	3	60	69
MO	6	3	60	69
actb	60	30	600	690
hd	20	10	200	230
hdk	60	30	600	690
pctb	60	30	600	690
popsh	2	0	20	22
Total	220	109	2,200	2,529

```
. li measure b se if inlist(measure,"MO","H","A") & loa == "nat" & k == 33 , noo
> b
```

measure	b	se
H	0.3352	0.0055
MO	0.1424	0.0025
A	0.4248	0.0019

```
. recode subg (0=0 "rural") (1=1 "urban") if loa == "area" , gen(area)
(0 differences between subg and area)

. lab var area area

. tabdisp indicator measure area if inlist(measure,"hd","hdk") ///
>      & !mi(area) & inlist(k,33,.) , cell(b)
```

indicator	area and measure			
	rural		urban	
	hd	hdk	hd	hdk
d_asst	0.2795	0.1258	0.2856	0.1235
d_ckfl	0.1540	0.0732	0.1437	0.0624
d_cm	0.0681	0.0527	0.0595	0.0422
d_educ	0.2980	0.1872	0.3002	0.1895
d_elct	0.4007	0.1688	0.3954	0.1668
d_hsg	0.4044	0.1604	0.4273	0.1805
d_nutr	0.2416	0.1617	0.2595	0.1720
d_sani	0.2584	0.1134	0.2259	0.0943
d_satt	0.3208	0.2057	0.3148	0.1935
d_wtr	0.2728	0.1285	0.2744	0.1251

## Example 2: Avoiding unnecessary estimations

```

. mkdir results

. use syn_cdt.a.dta if t == 1 , clear
. svyset psu [pw=weight], strata(stratum)
  (output omitted)
. mpitb set , name(trial01) desc(preferred spec) ///
>           d1(d_cm d_nutr, name(hl)) ///
>           d2(d_satt d_educ, name(ed)) ///
>           d3(d_elct d_wtr d_sani d_hsg d_ckfl d_asst, name(ls))
. mpitb est , name(trial01) measures(all) indmeas(all) aux(hd) svy ///
>           k(1 10 20 33 40 (10) 100) over(region area, k(20 33 50) indk(30)) ///
>           indk(20 33 40) weight(equal) lsa(results/trial01, replace)
  (output omitted)
. describe using results/trial01 , s

Contains data
  Observations:           1,232                17 May 2022 15:37
    Variables:              14
Sorted by:

```

➔ 1,232 instead of 2,529 estimates (while having more  $k$  for national level)

## Example 3: Adding alternative weights and indicator selections

```
. mpitb est , n(trial01) m(all) k(33) w(dimw(.5 .25 .25) name(health50)) ///  
> lsa(results/health50, replace) svy  
(output omitted)  
  
. mpitb est , n(trial01) m(all) k(33) w(dimw(.25 .5 .25) name(educ50)) ///  
> lsa(results/educ50, replace) svy  
(output omitted)  
  
. mpitb est , n(trial01) m(all) k(33) w(dimw(.25 .25 .5) name(livst50)) ///  
> lsa(results/livstd50, replace) svy  
(output omitted)  
  
. mpitb est , n(trial01) m(all) k(33) lsa(results/ind_equal, replace) ///  
> w(indw(.1 .1 .1 .1 .1 .1 .1 .1 .1 .1)) name(ind_equal)) svy  
(output omitted)  
  
. mpitb set , n(trial02) d1(d_cm d_nutr, n(hl)) d2(d_satt d_educ, n(ed)) ///  
> d3(d_wtr d_sani d_hsg d_ckfl d_asst, name(ls)) desc(w/o electricity)  
(output omitted)  
  
. mpitb est , n(trial02) m(all) k(33) w(equal) svy ///  
> lsa(results/trial02, replace)  
(output omitted)
```



```
. clear
. save results/results , replace emptyok
. loc flist trial01 trial02 health50 educ50 livstd50 ind_equal
. foreach f in `flist' {
.     append using results/`f' , nol
. }
. save results/results , replace
```

## Example 4: Several years for a single country

**mpitb est**

```
mpitb est , name(mpiname) [ ...  
    cotmeasures(measures) cotoptions(olist)  
    cotklist(numlist) cotyear(varname)  
    cotframe(name [, sopts]) cotsave(filename [, sopts])  
    tvar(varname) ... ]
```

```
. use syn_cdt.a.dta , clear
. svyset psu [pw=weight], strata(stratum)
. mpitb set , name(trial01) desc(preferred spec) ///
>         d1(d_cm d_nutr, name(hl)) ///
>         d2(d_satt d_educ, name(ed)) ///
>         d3(d_elct d_wtr d_sani d_hsg d_ckfl d_asst, name(ls))

. mpitb est , name(trial01) measures(all) klist(1 33 50) weight(equal) ///
>         lframe(myresults, replace) svy over(region) ///
>         cotmeasures(MO H A) cotframe(mycot, replace) tvar(t) cotyear(year)
(output omitted)
```


```
. frame myresults : sort t k
. frame myresults : li measure wgts t k b se if measure == "H" & loa == "nat" //
> /
> , noobs sepby(t)
```

measure	wgts	t	k	b	se
H	equal	1	1	0.9575	0.0024
H	equal	1	33	0.3352	0.0055
H	equal	1	50	0.0818	0.0032
H	equal	2	1	0.9205	0.0030
H	equal	2	33	0.2308	0.0047
H	equal	2	50	0.0411	0.0023

```
. frame mycot : li measure wgts ann t0 t1 k ctype b se if measure == "H" ///
> & loa == "nat" & ann == 0 , noobs sepby(k)
```

measure	wgts	ann	t0	t1	k	ctype	b	se
H	equal	0	1	2	1	abs	-0.0370	0.0038
H	equal	0	1	2	1	rel	-3.8665	0.3960
H	equal	0	1	2	33	abs	-0.1044	0.0074
H	equal	0	1	2	33	rel	-31.1428	1.8322
H	equal	0	1	2	50	abs	-0.0407	0.0040
H	equal	0	1	2	50	rel	-49.7432	3.4710

## Example 5: A single year for several countries



**mpitb refsh**

```
mpitb refsh using filename , id(name) [ ,  
  path(string) file(filename)  
  clear newfiles update(clist)  
  sid(sid) keep(namelist) char(clist)  
  depind(depind) gentvar(year) ]
```



```
. dir cdta , wide
syn_ABC_cdta.dta  syn_DEF_cdta.dta  syn_GHI_cdta.dta

. clear all
. mpitb refsh using results/refsh.dta, clear id(ccty) sid(region) p(cdta) ///
>      char(ccty cnum survey year cty)
(output omitted)
. li ccty region region_name survey year fname in 1/5, noob sepby(ccty)
```

ccty	region	region_name	survey	year	fname
DEF	3	DEF - region 3	MICS	2018-2019	syn_DEF_cdta
DEF	8	DEF - region 8	MICS	2018-2019	syn_DEF_cdta
DEF	11	DEF - region 11	MICS	2018-2019	syn_DEF_cdta
DEF	15	DEF - region 15	MICS	2018-2019	syn_DEF_cdta
DEF	2	DEF - region 2	MICS	2018-2019	syn_DEF_cdta

```
mpitb ctyselect
```

```
mpitb ctyselect varname [ if ] [ in ] [ ,  
    select(ctylist) rexp(regex) ]
```

```

. mkf rs
. frame rs: use results/refsh.dta , clear

. frame rs: mpitb ctyselect ccty
. foreach cty in `r(ctylist)' {
.     frame rs : qui levelsof fname if ccty == "`cty'" , loc(fname) clean
.     use `"cdata/`fname'" , clear
.     svyset psu [pw=weight] , strata(stratum) singleunit(centered)
.
.     mpitb set , n(mympi) d1(d_cm d_nutr, n(hl)) ///
>         d2(d_satt d_educ, n(ed)) ///
>         d3(d_elct d_wtr d_sani d_hsg d_ckfl d_asst, name(ls))
.
.     mpitb est , name(mympi) measures(all) klist(33) weight(equal) ///
>         lsa(results/`cty'_results, replace) over(region) ///
>         svy addmeta(ccty=`cty')
. }

```

```
. clear
. save results/results , replace emptyok
. loc flist : dir "results/" files "*_results.dta"
. foreach f in `flist' {
.     append using results/`f' , nol
. }

. gen region = subg if loa == "region"
. frlink m:1 ccty region , frame(rs)
. frget region_name , from(rs)
. save results/results.dta , replace
```

(and using the reference sheet again!)

```
. tabdisp ccty measure if loa == "nat" & inlist(k,33,.), cell(b)
```

ccty	measure		
	A	H	MO
ABC	0.4248	0.3352	0.1424
DEF	0.4070	0.2308	0.0940
GHI	0.4070	0.2308	0.0940

```
. tabdisp region_name measure if loa == "region" & inlist(k,33,.) & ccty == "ABC"
> " , c(b) 1
```

name in c-data	measure			
	A	H	M0	popsh
ABC - region 1	0.4264	0.3654	0.1558	0.0545
ABC - region 10	0.4337	0.3764	0.1632	0.0479
ABC - region 11	0.4359	0.3079	0.1342	0.0511
ABC - region 12	0.4187	0.3235	0.1355	0.0503
ABC - region 13	0.4226	0.3103	0.1312	0.0507
ABC - region 14	0.4253	0.3536	0.1504	0.0487
ABC - region 15	0.4289	0.3198	0.1372	0.0496
ABC - region 16	0.4160	0.3362	0.1398	0.0468
ABC - region 17	0.4211	0.3190	0.1343	0.0531
ABC - region 18	0.4261	0.2882	0.1228	0.0537
ABC - region 19	0.4125	0.3631	0.1498	0.0496
ABC - region 2	0.4314	0.3162	0.1364	0.0472
ABC - region 20	0.4255	0.3256	0.1385	0.0521
ABC - region 3	0.4299	0.3500	0.1505	0.0484
ABC - region 4	0.4367	0.3010	0.1315	0.0514
ABC - region 5	0.4223	0.3432	0.1449	0.0502
ABC - region 6	0.4261	0.3664	0.1561	0.0488
ABC - region 7	0.4178	0.3220	0.1345	0.0514
ABC - region 8	0.4176	0.3380	0.1411	0.0486
ABC - region 9	0.4240	0.3900	0.1654	0.0459




Paper available under

 <https://ophi.org.uk/rp-62a/>

Questions, comments, and suggestions are always welcome under

 [nsuppa@ced.uab.es](mailto:nsuppa@ced.uab.es)

 [@nicolaisuppa](#) | [@CEDemografia](#) | [@ophi\\_oxford](#)

-  Alkire, Sabina and James Foster (2011). “Counting and Multidimensional Poverty Measurement”. **Journal of Public Economics** 95.7-8, pp. 476–487. url: <https://www.sciencedirect.com/science/article/abs/pii/S0047272710001660>.
-  Alkire, Sabina, James Foster, et al. (2015). **Multidimensional Poverty Measurement and Analysis: A Counting Approach**. Oxford: Oxford University Press. url: <https://multidimensionalpoverty.org/>.
-  Suppa, Nicolai (2021). **The production process of the Global MPI**. Presentation at the 2021 UK Stata conference. url: <https://www.stata.com/meeting/uk21/>.